Pensieve header: Finding the MVA in β-calculus, with Oleg Chterental.

```
<< KnotTheory`
```

```
Loading KnotTheory` version of February 5, 2013, 3:48:46.4762.
Read more at http://katlas.org/wiki/KnotTheory.
```

```
β = B[hs, ts, ω, Λ]
```

```
βSimplify[B[hs_, ts_, ω_, Λ_]] := B[hs, ts, Factor[ω], Sum[
    Factor[Coefficient[Λ, t[u] h[x]]] * t[u] h[x],
    {u, ts}, {x, hs}
   ]];
βForm[B[hs_, ts_, ω_, Λ_]] := Module[{mat},
   mat = Table[
     Coefficient[Λ, t[u] h[x]],
     {u, ts}, {x, hs}
    ];
   PrependTo[mat, hs];
   mat = Transpose[Prepend[Transpose[mat], Prepend[ts, ω]]];
   MatrixForm[mat]
  ];
Format[β_B, StandardForm] := βForm[β]
```

```
B /: B[hs1_, ts1_, ω1_, Λ1_] B[hs2_, ts2_, ω2_, Λ2_] :=
  B[Union[hs1, hs2], Union[ts1, ts2], ω1 * ω2, Λ1 + Λ2] // βSimplify;
tm[u_, v_, w_][B[hs_, ts_, ω_, Λ_]] := B[
    hs, Union[Complement[ts, {u, v}], {w}],
    ω /. {t[u | v] → t[w], T[u | v] → T[w]},
    Λ /. {t[u | v] → t[w], T[u | v] → T[w]}
   ] // βSimplify;
⟨μ_⟩ := μ /. t[u_] → 1;
hm[x_, y_, z_][B[hs_, ts_, ω_, Λ_]] := Module[{α, β, γ},
   α = Coefficient[Λ, h[x]];
   β = Coefficient[Λ, h[y]];
   γ = Λ /. h[x | y] → 0;
   B[
     Union[Complement[hs, {x, y}], {z}], ts,
     ω, (α + β + ⟨α⟩ β) h[z] + γ
    ] // βSimplify
  ];
swap[u_, x_][B[hs_, ts_, ω_, Λ_]] := Module[{α, β, γ, δ, ϵ},
   α = Coefficient[Λ, t[u] h[x]];
   β = Coefficient[Λ /. h[x] → 0, t[u]];
   γ = Coefficient[Λ /. t[u] → 0, h[x]];
   δ = Λ /. {t[u] → 0, h[x] → 0};
   ϵ = 1 + α;
   B[hs, ts, ω ϵ,
     α (1 + ⟨γ⟩ / ϵ) t[u] h[x] + β (1 + ⟨γ⟩ / ϵ) t[u]
      + (γ / ϵ) h[x] + δ - γ β / ϵ
    ] // βSimplify
  ];
gm[a_, b_, c_][β_] := β // swap[a, b] // tm[a, b, a] // hm[a, b, a];
Rp[a_, b_] := B[{a, b}, {a, b}, 1, t[a] h[b] (T[a] - 1)];
Rm[a_, b_] := B[{a, b}, {a, b}, 1, t[a] h[b] (1 / T[a] - 1)];

βZ[L_] := Module[{s, β},
  s = Skeleton[L];
  β = Times @@ PD[L] /. X[i_, j_, k_, l_] :> If[PositiveQ[X[i, j, k, l]],
      Rp[l, i], Rm[j, i]
     ];
  Do[
   β = β // gm[s[[c, 1]], s[[c, k]], s[[c, 1]]],
   {c, 1, Length[s]}, {k, 2, Length[s[[c]]]}
  ];
  β
 ]
```

β = B[{1, 2}, {1, 2, 3, 4}, ω, Sum[α[i, j] t[i] h[j], {i, 1, 4}, {j, 1, 2}]]

ω	1	2
1	α[1, 1]	α[1, 2]
2	α[2, 1]	α[2, 2]
3	α[3, 1]	α[3, 2]
4	α[4, 1]	α[4, 2]

β // tm[1, 2, 1]

ω	1	2
1	α[1, 1] + α[2, 1]	α[1, 2] + α[2, 2]
3	α[3, 1]	α[3, 2]
4	α[4, 1]	α[4, 2]

β // tm[1, 2, 1] // tm[1, 3, 1]

ω	1	2
1	α[1, 1] + α[2, 1] + α[3, 1]	α[1, 2] + α[2, 2] + α[3, 2]
4	α[4, 1]	α[4, 2]

β // tm[2, 3, 2] // tm[1, 2, 1]

ω	1	2
1	α[1, 1] + α[2, 1] + α[3, 1]	α[1, 2] + α[2, 2] + α[3, 2]
4	α[4, 1]	α[4, 2]

β = B[{1, 2, 3, 4}, {1, 2}, ω, Sum[α[i, j] t[i] h[j], {i, 1, 2}, {j, 1, 4}]]

ω	1	2	3	4
1	α[1, 1]	α[1, 2]	α[1, 3]	α[1, 4]
2	α[2, 1]	α[2, 2]	α[2, 3]	α[2, 4]

β // hm[1, 2, 1]

ω	1	3	4
1	α[1, 1] + α[1, 2] + α[1, 1] α[1, 2] + α[1, 2] α[2, 1]	α[1, 3]	α[1, 4]
2	α[2, 1] + α[2, 2] + α[1, 1] α[2, 2] + α[2, 1] α[2, 2]	α[2, 3]	α[2, 4]

β // hm[1, 2, 1] // hm[1, 3, 1]

ω	
1	α[1, 1] + α[1, 2] + α[1, 1] α[1, 2] + α[1, 3] + α[1, 1] α[1, 3] + α[1, 2] α[1, 3] + α[1, 1] α[1,
2	α[2, 1] + α[2, 2] + α[1, 1] α[2, 2] + α[2, 1] α[2, 2] + α[2, 3] + α[1, 1] α[2, 3] + α[1, 2] α[2,

β // hm[2, 3, 2] // hm[1, 2, 1]

ω	
1	α[1, 1] + α[1, 2] + α[1, 1] α[1, 2] + α[1, 3] + α[1, 1] α[1, 3] + α[1, 2] α[1, 3] + α[1, 1] α[1,
2	α[2, 1] + α[2, 2] + α[1, 1] α[2, 2] + α[2, 1] α[2, 2] + α[2, 3] + α[1, 1] α[2, 3] + α[1, 2] α[2,

(β // hm[1, 2, 1] // hm[1, 3, 1]) == (β // hm[2, 3, 2] // hm[1, 2, 1])

True

```
β = B[{1, 2}, {1, 2, 3}, ω, Sum[α[i, j] t[i] h[j], {i, 1, 3}, {j, 1, 2}]]
```

$$\begin{pmatrix} \omega & 1 & 2 \\ 1 & \alpha[1,1] & \alpha[1,2] \\ 2 & \alpha[2,1] & \alpha[2,2] \\ 3 & \alpha[3,1] & \alpha[3,2] \end{pmatrix}$$

```
O1 = β // tm[1, 2, 1] // swap[1, 1]
```

$$\begin{pmatrix} \omega\,(1+\alpha[1,1]+\alpha[2,1]) & 1 & 2 \\ 1 & \frac{(\alpha[1,1]+\alpha[2,1])\,(1+\alpha[1,1]+\alpha[2,1]+\alpha[3,1])}{1+\alpha[1,1]+\alpha[2,1]} & \frac{(\alpha[1,2]+\alpha[2,2])\,(1+\alpha[1,1]+\alpha[2,}{1+\alpha[1,1]+\alpha[2,1]} \\ 3 & \frac{\alpha[3,1]}{1+\alpha[1,1]+\alpha[2,1]} & \frac{-\alpha[1,2]\,\alpha[3,1]-\alpha[2,2]\,\alpha[3,1]+\alpha[3,2]+\alpha[1,1}{1+\alpha[1,1]+\alpha[2,1]} \end{pmatrix}$$

```
O2 = β // swap[2, 1] // swap[1, 1] // tm[1, 2, 1]
```

$$\begin{pmatrix} \omega\,(1+\alpha[1,1]+\alpha[2,1]) & 1 & 2 \\ 1 & \frac{(\alpha[1,1]+\alpha[2,1])\,(1+\alpha[1,1]+\alpha[2,1]+\alpha[3,1])}{1+\alpha[1,1]+\alpha[2,1]} & \frac{(\alpha[1,2]+\alpha[2,2])\,(1+\alpha[1,1]+\alpha[2,}{1+\alpha[1,1]+\alpha[2,1]} \\ 3 & \frac{\alpha[3,1]}{1+\alpha[1,1]+\alpha[2,1]} & \frac{-\alpha[1,2]\,\alpha[3,1]-\alpha[2,2]\,\alpha[3,1]+\alpha[3,2]+\alpha[1,1}{1+\alpha[1,1]+\alpha[2,1]} \end{pmatrix}$$

```
O1 == O2
```

True

```
β = B[{1, 2, 3}, {1, 2}, ω, Sum[α_{i,j} t[i] h[j], {i, 1, 2}, {j, 1, 3}]]
```

$$\begin{pmatrix} \omega & 1 & 2 & 3 \\ 1 & \alpha_{1,1} & \alpha_{1,2} & \alpha_{1,3} \\ 2 & \alpha_{2,1} & \alpha_{2,2} & \alpha_{2,3} \end{pmatrix}$$

```
O1 = β // hm[1, 2, 1] // swap[1, 1]
```

$$\begin{pmatrix} \omega\,(1+\alpha_{1,1}+\alpha_{1,2}+\alpha_{1,1}\,\alpha_{1,2}+\alpha_{1,2}\,\alpha_{2,1}) & 1 & \\ 1 & \frac{(1+\alpha_{1,1}+\alpha_{2,1})\,(\alpha_{1,1}+\alpha_{1,2}+\alpha_{1,1}\,\alpha_{1,2}+\alpha_{1,2}\,\alpha_{2,1})\,(1+\alpha_{1,2}+\alpha_{2,2})}{1+\alpha_{1,1}+\alpha_{1,2}+\alpha_{1,1}\,\alpha_{1,2}+\alpha_{1,2}\,\alpha_{2,1}} & \\ 2 & \frac{\alpha_{2,1}+\alpha_{2,2}+\alpha_{1,1}\,\alpha_{2,2}+\alpha_{2,1}\,\alpha_{2,2}}{1+\alpha_{1,1}+\alpha_{1,2}+\alpha_{1,1}\,\alpha_{1,2}+\alpha_{1,2}\,\alpha_{2,1}} & -\frac{\alpha_{1,3}\,\alpha_{2,1}+\alpha_{1,3}\,\alpha_{2,}}{} \end{pmatrix}$$

```
O2 = β // swap[1, 1] // swap[1, 2] // hm[1, 2, 1]
```

$$\begin{pmatrix} \omega\,(1+\alpha_{1,1}+\alpha_{1,2}+\alpha_{1,1}\,\alpha_{1,2}+\alpha_{1,2}\,\alpha_{2,1}) & 1 & \\ 1 & \frac{(1+\alpha_{1,1}+\alpha_{2,1})\,(\alpha_{1,1}+\alpha_{1,2}+\alpha_{1,1}\,\alpha_{1,2}+\alpha_{1,2}\,\alpha_{2,1})\,(1+\alpha_{1,2}+\alpha_{2,2})}{1+\alpha_{1,1}+\alpha_{1,2}+\alpha_{1,1}\,\alpha_{1,2}+\alpha_{1,2}\,\alpha_{2,1}} & \\ 2 & \frac{\alpha_{2,1}+\alpha_{2,2}+\alpha_{1,1}\,\alpha_{2,2}+\alpha_{2,1}\,\alpha_{2,2}}{1+\alpha_{1,1}+\alpha_{1,2}+\alpha_{1,1}\,\alpha_{1,2}+\alpha_{1,2}\,\alpha_{2,1}} & \frac{-\alpha_{1,3}\,\alpha_{2,1}-\alpha_{1,3}\,\alpha_{2,2}}{} \end{pmatrix}$$

```
O1 == O2
```

$$\begin{pmatrix} \omega\,(1+\alpha_{1,1}+\alpha_{1,2}+\alpha_{1,1}\,\alpha_{1,2}+\alpha_{1,2}\,\alpha_{2,1}) & 1 & \\ 1 & \frac{(1+\alpha_{1,1}+\alpha_{2,1})\,(\alpha_{1,1}+\alpha_{1,2}+\alpha_{1,1}\,\alpha_{1,2}+\alpha_{1,2}\,\alpha_{2,1})\,(1+\alpha_{1,2}+\alpha_{2,2})}{1+\alpha_{1,1}+\alpha_{1,2}+\alpha_{1,1}\,\alpha_{1,2}+\alpha_{1,2}\,\alpha_{2,1}} & \\ 2 & \frac{\alpha_{2,1}+\alpha_{2,2}+\alpha_{1,1}\,\alpha_{2,2}+\alpha_{2,1}\,\alpha_{2,2}}{1+\alpha_{1,1}+\alpha_{1,2}+\alpha_{1,1}\,\alpha_{1,2}+\alpha_{1,2}\,\alpha_{2,1}} & -\frac{\alpha_{1,3}\,\alpha_{2,1}+\alpha_{1,3}\,\alpha_{2,}}{} \end{pmatrix}$$

```
β = B[{1, 2, 3, 4}, {1, 2, 3, 4}, ω, Sum[α_{i,j} t[i] h[j], {i, 1, 4}, {j, 1, 4}]]
```

$$\begin{pmatrix} \omega & 1 & 2 & 3 & 4 \\ 1 & \alpha_{1,1} & \alpha_{1,2} & \alpha_{1,3} & \alpha_{1,4} \\ 2 & \alpha_{2,1} & \alpha_{2,2} & \alpha_{2,3} & \alpha_{2,4} \\ 3 & \alpha_{3,1} & \alpha_{3,2} & \alpha_{3,3} & \alpha_{3,4} \\ 4 & \alpha_{4,1} & \alpha_{4,2} & \alpha_{4,3} & \alpha_{4,4} \end{pmatrix}$$

```
O1 = β // gm[1, 2, 1] // gm[1, 3, 1]
```

A very large output was generated. Here is a sample of it:

$$\begin{pmatrix} \omega\,(1+\alpha_{1,2}+\alpha_{1,3}+\alpha_{1,2}\,\alpha_{1,3}+\alpha_{2,3}+\alpha_{1,2}\,\alpha_{2,3}+\alpha_{1,3}\,\alpha_{3,2}+\alpha_{1,3}\,\alpha_{4,2}) & 1 \\ 1 & \frac{\alpha_{1,1}+\alpha_{1,2}+\ll 645\gg+\alpha_{1,3}\,\alpha}{1+\alpha_{1,2}+\alpha_{1,3}+\ll 1\gg\ \ll 1\gg\ \ll 1\gg+\alpha_{1}\ \ll 1\gg\ \ll 1} \\ 4 & \frac{\ll 1\gg}{\ll 1\gg} \end{pmatrix}$$

Show Less | Show More | Show Full Output | Set Size Limit...

```
O2 = β // gm[2, 3, 2] // gm[1, 2, 1]
```

A very large output was generated. Here is a sample of it:

$$\begin{pmatrix} \omega\,(1+\alpha_{1,2}+\alpha_{1,3}+\alpha_{1,2}\,\alpha_{1,3}+\alpha_{2,3}+\alpha_{1,2}\,\alpha_{2,3}+\alpha_{1,3}\,\alpha_{3,2}+\alpha_{1,3}\,\alpha_{4,2}) & 1 \\ 1 & \frac{\alpha_{1,1}+\alpha_{1,2}+\ll 645\gg+\alpha_{1,3}\,\alpha}{1+\alpha_{1,2}+\alpha_{1,3}+\ll 1\gg\ \ll 1\gg\ \ll 1\gg+\alpha_{1}\ \ll 1\gg\ \ll 1} \\ 4 & \frac{\ll 1\gg}{\ll 1\gg} \end{pmatrix}$$

Show Less | Show More | Show Full Output | Set Size Limit...

```
O1 == O2
```

```
True
```

```
L = Knot["8_16"]
```

```
Knot[8, 16]
```

DrawMorseLink[L]

KnotTheory::loading : Loading precomputed data in PD4Knots`.

KnotTheory::credits : MorseLink was added to KnotTheory` by Siddarth Sankaran at the University of Toronto in the summer of 2005.

KnotTheory::credits : DrawMorseLink was written by Siddarth Sankaran at the University of Toronto in the summer of 2005.



PD[L]

PD[X[6, 2, 7, 1], X[14, 6, 15, 5], X[16, 11, 1, 12], X[12, 7, 13, 8], X[8, 3, 9, 4], X[4, 9, 5, 10], X[10, 15, 11, 16], X[2, 14, 3, 13]]

βZ[Knot[4, 1]]

$$\begin{pmatrix} -1+3\,T[1]-T[1]^2 & 1 \\ 1 & 0 \end{pmatrix}$$

s = Skeleton[L]

{Loop[1, 2, 3, 4, 5, 6, 7, 8, 9, 10, 11, 12, 13, 14, 15, 16]}

```
β = Times @@ PD[L] /. X[i_, j_, k_, l_] :> If[PositiveQ[X[i, j, k, l]],
    Rp[l, i], Rm[j, i]
   ]
```

1	1	2	3	4	5	6	7	8	9	10	11	12	13	14
1	0	0	0	0	0	$-1+T[1]$	0	0	0	0	0	0	0	0
2	0	0	0	0	0	0	0	0	0	0	0	0	0	0
3	0	0	0	0	0	0	0	$-\frac{-1+T[3]}{T[3]}$	0	0	0	0	0	0
4	0	0	0	0	0	0	0	0	0	0	0	0	0	0
5	0	0	0	0	0	0	0	0	0	0	0	0	0	$-1+T[5]$
6	0	0	0	0	0	0	0	0	0	0	0	0	0	0
7	0	0	0	0	0	0	0	0	0	0	0	$-\frac{-1+T[7]}{T[7]}$	0	0
8	0	0	0	0	0	0	0	0	0	0	0	0	0	0
9	0	0	0	$-\frac{-1+T[9]}{T[9]}$	0	0	0	0	0	0	0	0	0	0
10	0	0	0	0	0	0	0	0	0	0	0	0	0	0
11	0	0	0	0	0	0	0	0	0	0	0	0	0	0
12	0	0	0	0	0	0	0	0	0	0	0	0	0	0
13	0	$-1+T[13]$	0	0	0	0	0	0	0	0	0	0	0	0
14	0	0	0	0	0	0	0	0	0	0	0	0	0	0
15	0	0	0	0	0	0	0	0	0	$-\frac{-1+T[15]}{T[15]}$	0	0	0	0
16	0	0	0	0	0	0	0	0	0	0	0	0	0	0

```
Do[
 β = β // gm[s[[c, 1]], s[[c, k]], s[[c, 1]]],
 {c, 1, Length[s]}, {k, 2, Length[s[[c]]]}
]
```

```
β
```

$$\begin{pmatrix} \frac{1-4\,T[1]+8\,T[1]^2-9\,T[1]^3+8\,T[1]^4-4\,T[1]^5+T[1]^6}{T[1]^4} & 1 \\ 1 & -\frac{(-1+T[1])\,(1+T[1])}{T[1]^2} \end{pmatrix}$$

```
Alexander[L][T[1]]
```

$$-9+\frac{1}{T[1]^3}-\frac{4}{T[1]^2}+\frac{8}{T[1]}+8\,T[1]-4\,T[1]^2+T[1]^3$$

```
AllKnots[{3, 7}]
```

{Knot[3, 1], Knot[4, 1], Knot[5, 1], Knot[5, 2], Knot[6, 1], Knot[6, 2], Knot[6, 3], Knot[7, 1], Knot[7, 2], Knot[7, 3], Knot[7, 4], Knot[7, 5], Knot[7, 6], Knot[7, 7]}

```
Table[
 Factor[βZ[L][[3]] / Alexander[L][T[1]]],
 {L, AllKnots[{3, 8}]}
]
```

```
{1/T[1], T[1], 1/T[1]^2, 1/T[1]^2, 1, 1, 1, 1/T[1]^3, 1/T[1]^3, T[1]^4, T[1]^4, 1/T[1]^3,
 1/T[1], T[1]^2, 1/T[1], 1/T[1], T[1], T[1], T[1]^3, 1/T[1], T[1], T[1], T[1],
 T[1], 1/T[1], T[1], T[1], 1/T[1], 1/T[1]^3, 1/T[1], T[1], 1, T[1]^4, 1, 1/T[1]}
```

```
βZ[Link["L6a4"]]
```

KnotTheory::loading : Loading precomputed data in PD4Links`.

```
( (1-T[1]-T[5]+T[1] T[5]-T[9]+T[1] T[9]+T[5] T[9]) (T[1]+T[5]-T[1] T[5]+T[9]-T[1] T[9]-T[5] T[9]+T[1] T[5] T[9])
  ---------------------------------------------------------------------------------------------------------------
                                                 T[1] T[5] T[9]
                                                  1                                                          (1-T
                                                  5                                                        - (1-'
                                                  9                                                        - (1-'
```

```
L = Link["L6a4"]
```

```
Link[6, Alternating, 4]
```

```
Skeleton[L]
```

```
{Loop[1, 2, 3, 4], Loop[5, 6, 7, 8], Loop[9, 10, 11, 12]}
```

```
βMVA0[L_] := Module[{hs, ts, ω, Λ, gs, mat, res},
  {hs, ts, ω, Λ} = List @@ βZ[L];
  gs = First /@ Skeleton[L];
  mat = Table[
    Coefficient[Λ - (⟨Λ⟩ /. h[a_] ⧴ t[a] h[a]), t[u] h[x]],
    {u, gs // Rest}, {x, gs // Rest}
   ];
  res = ω Det[mat] / (T[Skeleton[L][[1, 1]]] - 1) // Factor;
  res /. T[k_] ⧴ T[Position[gs, k][[1, 1]]]
 ]
```

```
βtr[B[hs_, ts_, ω_, Λ_]] := Module[{mat, res},
   mat = Table[
     Coefficient[Λ - (⟨Λ⟩ /. h[a_] ⧴ t[a] h[a]), t[u] h[x]],
     {u, hs // Rest}, {x, hs // Rest}
    ];
   res = ω Det[mat] / (T[ts[[1]]] - 1) // Factor;
   res /. T[k_] ⧴ T[Position[hs, k][[1, 1]]]
  ];
βMVA[L_] := βtr[βZ[L]]
```

βMVA[L]

$$\frac{(-1+T[1])\,(-1+T[2])\,(-1+T[3])}{T[1]\,T[2]}$$

MultivariableAlexander[L][T]

KnotTheory::loading : Loading precomputed data in MultivariableAlexander4Links`.

$$\frac{(-1+T[1])\,(-1+T[2])\,(-1+T[3])}{\sqrt{T[1]}\,\sqrt{T[2]}\,\sqrt{T[3]}}$$

```
Table[
 Factor[βMVA[L] / MultivariableAlexander[L][T]],
 {L, AllLinks[{2, 7}]}
]
```

$$\Big\{-\frac{1}{T[1]^2\,T[2]},\ -\frac{1}{T[1]^{3/2}\,T[2]^{3/2}},\ -\frac{1}{\sqrt{T[1]}\,T[2]^{3/2}},\ -\frac{1}{T[1]^{3/2}\,\sqrt{T[2]}},$$
$$-\frac{1}{T[1]^2\,T[2]^2},\ -\frac{1}{T[1]^2\,T[2]^2},\ \frac{\sqrt{T[3]}}{\sqrt{T[1]}\,\sqrt{T[2]}},\ \frac{1}{T[1]^{3/2}\,T[2]^{3/2}\,T[3]^{3/2}},$$
$$\frac{T[3]^{3/2}}{\sqrt{T[1]}\,\sqrt{T[2]}},\ -\frac{1}{\sqrt{T[1]}\,\sqrt{T[2]}},\ -\frac{1}{T[1]^{3/2}\,T[2]^{7/2}},\ -\frac{T[2]^{3/2}}{\sqrt{T[1]}},\ -\frac{T[2]^{3/2}}{\sqrt{T[1]}},$$
$$-\frac{1}{T[1]\,T[2]^2},\ -T[2],\ \frac{\sqrt{T[3]}}{T[1]^{3/2}\,\sqrt{T[2]}},\ -\frac{1}{T[1]^{3/2}\,T[2]^{7/2}},\ -\frac{1}{\sqrt{T[1]}\,T[2]^{5/2}}\Big\}$$

β = B[{1, 2, 3}, {1, 2, 3}, ω, Sum[α[i, j] t[i] h[j], {i, 1, 3}, {j, 1, 3}]]

$$\begin{pmatrix} \omega & 1 & 2 & 3 \\ 1 & \alpha[1,1] & \alpha[1,2] & \alpha[1,3] \\ 2 & \alpha[2,1] & \alpha[2,2] & \alpha[2,3] \\ 3 & \alpha[3,1] & \alpha[3,2] & \alpha[3,3] \end{pmatrix}$$

β // gm[2, 3, 2]

$$\begin{pmatrix} \omega\,(1+\alpha[2,3]) & 1 & \\ 1 & \frac{\alpha[1,1]-\alpha[1,3]\,\alpha[2,1]+\alpha[1,1]\,\alpha[2,3]}{1+\alpha[2,3]} & \\ 2 & \frac{\alpha[2,1]+\alpha[1,3]\,\alpha[2,1]+\alpha[2,1]\,\alpha[2,3]+\alpha[3,1]+\alpha[2,3]\,\alpha[3,1]}{1+\alpha[2,3]} & \alpha[2,2]+\alpha[1,3]\,\alpha[2,2]+\alpha[2,3]+\alpha[1,2]\,\alpha[2, \end{pmatrix}$$

β // gm[2, 3, 2] // βtr

$$-\frac{1}{-1+T[1]}\omega\,(\alpha[1,2]+\alpha[1,3]+\alpha[1,2]\,\alpha[1,3]+\alpha[1,2]\,\alpha[2,3]+\alpha[1,3]\,\alpha[3,2])$$

β // gm[3, 2, 2]

$$\begin{pmatrix} \omega\,(1+\alpha[3,2]) & 1 & \\ 1 & \frac{\alpha[1,1]-\alpha[1,2]\,\alpha[3,1]+\alpha[1,1]\,\alpha[3,2]}{1+\alpha[3,2]} & \\ 3 & \frac{\alpha[2,1]+\alpha[3,1]+\alpha[1,2]\,\alpha[3,1]+\alpha[2,1]\,\alpha[3,2]+\alpha[3,1]\,\alpha[3,2]}{1+\alpha[3,2]} & \alpha[2,2]+\alpha[1,3]\,\alpha[2,2]+\alpha[2,3]+\alpha[2,2]\,\alpha[2, \end{pmatrix}$$

```
β // gm[3, 2, 2] // βtr
```

$$-\frac{1}{-1+T[1]}\omega\,(\alpha[1,2]+\alpha[1,3]+\alpha[1,2]\,\alpha[1,3]+\alpha[1,2]\,\alpha[2,3]+\alpha[1,3]\,\alpha[3,2])$$

```
(β // gm[2, 3, 2] // βtr) == (β // gm[3, 2, 2] // βtr)
```

True

```
n = 5;
β = B[Range[n], Range[n], ω, Sum[α[i, j] t[i] h[j], {i, 1, n}, {j, 1, n}]]
```

$$\begin{pmatrix} \omega & 1 & 2 & 3 & 4 & 5 \\ 1 & \alpha[1,1] & \alpha[1,2] & \alpha[1,3] & \alpha[1,4] & \alpha[1,5] \\ 2 & \alpha[2,1] & \alpha[2,2] & \alpha[2,3] & \alpha[2,4] & \alpha[2,5] \\ 3 & \alpha[3,1] & \alpha[3,2] & \alpha[3,3] & \alpha[3,4] & \alpha[3,5] \\ 4 & \alpha[4,1] & \alpha[4,2] & \alpha[4,3] & \alpha[4,4] & \alpha[4,5] \\ 5 & \alpha[5,1] & \alpha[5,2] & \alpha[5,3] & \alpha[5,4] & \alpha[5,5] \end{pmatrix}$$

```
(β // gm[2, 3, 2] // βtr) == (β // gm[3, 2, 2] // βtr)
```

True

```
(β // gm[2, 3, 2]) == (β // gm[3, 2, 2])
```

$$\begin{pmatrix} \omega\,(1+\alpha[2,3]) & 1 & \\ 1 & \frac{\alpha[1,1]-\alpha[1,3]\,\alpha[2,1]+\alpha[1,1]\,\alpha[2,3]}{1+\alpha[2,3]} & \\ 2 & \frac{\alpha[2,1]+\alpha[1,3]\,\alpha[2,1]+\alpha[2,1]\,\alpha[2,3]+\alpha[3,1]+\alpha[2,3]\,\alpha[3,1]+\alpha[2,1]\,\alpha[4,3]+\alpha[2,1]\,\alpha[5,3]}{1+\alpha[2,3]} & \frac{\alpha[2,2]+\alpha[1,}{} \\ 4 & \frac{\alpha[4,1]+\alpha[2,3]\,\alpha[4,1]-\alpha[2,1]\,\alpha[4,3]}{1+\alpha[2,3]} & \\ 5 & \frac{\alpha[5,1]+\alpha[2,3]\,\alpha[5,1]-\alpha[2,1]\,\alpha[5,3]}{1+\alpha[2,3]} & \end{pmatrix}$$

```
n = 2;
β = B[Range[n], Range[n], ω, Sum[α[i, j] t[i] h[j], {i, 1, n}, {j, 1, n}]]
```

$$\begin{pmatrix} \omega & 1 & 2 \\ 1 & \alpha[1,1] & \alpha[1,2] \\ 2 & \alpha[2,1] & \alpha[2,2] \end{pmatrix}$$

```
(β // gm[1, 2, 1] // βtr) == (β // gm[2, 1, 1] // βtr)
```

Det::matsq : Argument {} at position 1 is not a non−empty square matrix. ≫

Det::matsq : Argument {} at position 1 is not a non−empty square matrix. ≫

$$\frac{\omega\,\mathrm{Det}[\{\}]\,(1+\alpha[1,2])}{-1+T[1]} == \frac{\omega\,\mathrm{Det}[\{\}]\,(1+\alpha[2,1])}{-1+T[1]}$$

```
(β // gm[1, 2, 1]) == (β // gm[2, 1, 1])
```

$$\begin{pmatrix} \omega\,(1+\alpha[1,2]) & 1 \\ 1 & \alpha[1,1]+\alpha[1,2]+\alpha[1,1]\,\alpha[1,2]+\alpha[2,1]+\alpha[1,2]\,\alpha[2,1]+\alpha[2,2]+\alpha[1,1 \end{pmatrix}$$